






3. lecke – Mandalák



Bevezető – mik azok a mandalák?

A mandala eredetileg a buddhistavalláshoz kapcsolódik. A szó a szanszkrit nyelvből ered, a mandala szanszkrit nyelven „kör”-t jelent. Hagyományosan kör alakú alakra festett, geometriai alakzatokból álló, középpontosan szimmetrikus színes képről van szó, melyet a buddhisták a meditáció, a mély gondolkodás egy formájaként festettek. Festés közben az alkotó megnyugszik, elfeledkezik gondjairól, és elgondolkodhat a világ folyásáról és saját magáról. Így a mandala egyszerre jelképezi az alkotót magát, és az őt körülvevő világot, tükrözi alkotója lelkivilágát és a világhoz fűződő kapcsolatát.

Eredetileg tehát a kézzel festett mandalák minden egyes apró részletének külön jelentése volt. Manapság azonban sokan festenek vagy rajzolnak mandalát csak a szépsége miatt, mindenféle mögöttes jelentés nélkül. Mi is ezt fogjuk tenni, mégpedig a LOGO segítségével.

Hivatalos szabály tehát nem létezik arra, hogy mi is az a mandala. Általában viszont elmondhatjuk, hogy a mandala három legfontosabb tulajdonsága, hogy kör alapú, színes és középpontosan szimmetrikus legyen. Ennyit elég is tudni róla, kezdődhet a rajzolás!

Hogyan rajzoljunk LOGO-ban mandalát?


Mandaláink különböző apró, színes sokszögekből állnak majd, ezek a sokszögek mind egy-egy külön kis eljárást fognak jelenteni.

A sokszögekből először felépítjük a mandala egy „szeletét”, azután egy nagy körlapra ebből a szeletből körbe elhelyezünk annyit, amennyit csak szeretnénk. Majdnem úgy, mintha egy virágot rajzolnánk: először megrajzolunk egyetlen szírmot, aztán sokszor egymás mellé illesztjük ugyanazt a szírmot, amíg körbe nem érünk.

A sokszögek most egyik csúcson állnak majd, és ettől az alsó csúcstól kezdve rajzoljuk meg őket. Az egyes szeletek megrajzolása után pedig mindig vissza kell állni a szelet csúcsához, azaz a mandala közepéhez, hogy a következő szeletet is jó helyről kezdjük rajzolni.

A mandalák gyakori díszítőeleme a rombusz, írunk tehát először erre eljárást. A rombusz két paramétere az oldalhossz és az alsó és felső csúcsnál található szög lesz. Az ábra mutatja, hogy mikor mért annyit kell fordulni, amennyi az eljárásban szerepel.

```
tanuld rombusz :hossz :szög
  balra :szög/2 előre :hossz jobbra :szög előre :hossz
  jobbra 180-:szög előre :hossz
  jobbra :szög előre :hossz jobbra 180-:szög/2
vége
```





A díszítéshez bármilyen szabályos sokszöget is felhasználhatunk, úgyhogy írjunk egy olyan általános sokszög eljárást, amely csúcsára állított szabályos sokszöget rajzol, két paramétere pedig a csúcsok száma és az oldalhossz.

Ismétlésképpen nézzük meg, hogy nézett ki az általános sokszög eljárásunk:

```
tanuld soksz :n :hossz
ismétlés :n [előre :hossz jobbra 360/:n]
vége
```

Ezt az eljárást kell kiegészíteni azzal, hogy az elején forduljon el a teknőc a sokszög szögének felével, a végén pedig forduljon vissza (ugyanúgy, ahogy a rombusznál), így fog a sokszög a csúcsán állni. Kérdés, hogyan tudjuk kiszámolni a sokszög szögének felét?

Azt tudjuk, hogy a teknőcnek rajzolás közben annyit kell fordulnia, amennyi a külső szög, ez pedig $360^\circ/n$, ha n a sokszög csúcsainak száma. A belső szöget megkapjuk, ha 180° -ból kivonjuk a külső szöget, ez tehát $180^\circ - 360^\circ/n$. Ennek kell most a fele, ami tehát $(180^\circ - 360^\circ/n)/2$, ami egyszerűsítve $90^\circ - 180^\circ/n$. Ennyit kell tehát a teknőcnek elfordulnia balra az eljárás kezdete előtt. Az átalakított eljárásunk tehát a következő:

```
tanuld csúcssoksz :n :hossz
balra 90-180/:n
ismétlés :n [előre :hossz jobbra 360/:n]
jobbra 90-180/:n
vége
```

Írjunk egy kiszínező eljárást is, amely egy csúcsára állított sokszöget kiszínez. Mindössze annyit kell tennie, hogy felemelt tollal előre küldi pár lépéssel a teknőcot, színez, majd visszaáll eredeti helyére. Az eljárás paramétere a szín lesz, hogy milyenre színezzon, amit 0 és 15 közötti számmal tudunk majd megadni.

```
tanuld kitölt :szín
tollatfel előre 7 tollatle
töltőszín! :szín tölt
tollatfel hátra 7 tollatle
vége
```

Vigyázat! Az eljárás mindig az első határvonalig színez, tehát nem fog jól működni, ha egy-másba érő alakzatokat szeretnél kiszínezni, illetve túl apró alakzat esetén az egész képernyőt kiszínezheti. (Apró alakzatok esetén olyan színező eljárást használj, amely csak 2-3 lépést lép előre a színezéshez.)

Próbáld ki meglévő eljárásaidat az alábbiakhoz hasonló parancsokkal:

```
tollszín! 9
rombusz 100 30
kitölt 9
```

```
tollszín! 0
csúcssoksz 4 100
kitölt 4
csúcssoksz 3 100
kitölt 12
```

```
tollszín! 9
csúcssoksz 5 30
kitölt 11
```


```
tollszín! 5
rombusz 100 60
kitölt 5
tollszín! 3
csúcssoksz 3 100
kitölt 3
```

Színes körlapot pedig a már ismert **pont** parancs segítségével fogunk rajzolni, ahogyan a 2. leckében a koncentrikus köröket. A **pont** parancs használatánál arra kell csak figyelni, hogy előtte mindig jó nagyra kell állítani a tollvastagságot (attól függően, hogy mekkora kört szeretnénk rajzolni), utána azonban mindig állítsuk vissza 1-re, különben a további utasításokat is ezzel az elképesztő tollvastagsággal fogja végrehajtani.



Másra nincs is szükségünk az egyszerűbb mandalákhoz. Az első próbálkozásunk legyen olyan mandala, melynek díszítéséhez nincs más sokszögre szükségünk, csak rombuszokra.

Ennek egy szelete 4 darab rombuszból áll, a zöldből két kékéből és egy lilából. A rombuszok méretét és szögét, a közöttük lévő távolságokat és a színeket próbálgatással határoztam meg. Így tehát az eljárás:



Most már nincs is más dolgunk, minthogy rajzoljunk egy nagy körlapot (pontot), majd arra az **ismétlés** paranccsal körben sok ilyen szeletet. Ahhoz, hogy körbeérjünk a szeletekkel, ugyanazt a módszert kell alkalmaznunk, mint az 1. leckében a sokszögmintáknál, azaz az ismétlések száma szorozva a fordulási szöggel legyen mindig 360° . Az ábrán látható mandala megrajzolásához például 12 szeletet használtam fel, így két szelet megrajzolása között 30° -ot kell elfordulni:

```

tanuld mandala1
tollvastagság! 400 tollszín! 11 pont
tollvastagság! 1 ismétlés 12 [szelet1 jobbra 30]
vége

```

Próbáld ki a **mandala1** parancsot! Ugye, ugyanazt kapod, mint ami az ábrán látható?

Természetesen a színeket tetszőlegesen választhatod meg, bármilyen 0 és 15 közötti számot írhatasz a **tollszín!** és **kitölt** parancsok után. Sőt, az eljárásokat átírhatjuk úgy is, hogy a színeket paraméterekként lehessen megadni. Ekkor az eljárások így módosulnak:

```

tanuld szelet1param :szín1 :szín2 :szín3
tollszín! :szín1 rombusz 70 20 kitölt :szín1
tollatfel előre 150 tollatle
tollszín! :szín2 rombusz 20 30 kitölt :szín2
tollatfel hátra 20 balra 90 előre 10 tollatle
tollszín! :szín3 rombusz 30 150 kitölt :szín3
tollatfel jobbra 180 előre 20 tollatle
tollszín! :szín3 rombusz 30 150 kitölt :szín3
tollatfel hátra 10 balra 90 hátra 130 tollatle
vége

```

```


tanuld mandala1param :szín1 :szín2 :szín3 :szín4
tollvastagság! 400 tollszín! :szín4 pont
tollvastagság! 1
ismétlés 12 [szelet1param :szín1 :szín2 :szín3 jobbra 30]
vége

```


Így az előző mandalát a **mandala1param 2 5 9 11** paranccsal rajzolhatod meg, de más számokkal természetesen ugyanilyen mintájú, de más színű mandalákat kapsz. Az eljárást meghívhatod véletlenszámokkal is, ahogyan azt az 1. leckében a sokszögmintákkal tettük. Most négy darab szint kell paraméterként megadni, mind a négy helyett írhatunk **véletlenszám 16**-ot.



Tehát a **mandala1param véletlenszám 16 véletlenszám 16 véletlenszám 16 véletlenszám 16** parancs akárhányszor meghívod, annyszor más színű mandalát rajzol:



Természetesen nemcsak a színeket változtathatod, hanem azt is, hogy hány szelettel díszíted a mandalát. Ehhez a **mandala1** eljáráshoz kell átadni az ismétlések számát és az elfordulás szögét.



Érdekességek

